

Excel. Примеры использования функции ДВССЫЛ (INDIRECT)

Функция ДВССЫЛ (INDIRECT) — одна из наиболее трудных в освоении функций Excel. Однако умение использовать ее позволит вам решать многие из задач, кажущихся вам сейчас неразрешимыми. По сути, если в формуле есть раздел ДВССЫЛ со ссылкой на ячейку, эта ссылка обрабатывается как содержимое соответствующей ячейки.¹ Например (рис. 1), в ячейке C4 я ввел формулу =ДВССЫЛ(A4), и Excel возвратил значение, равное 6. Excel возвращает именно это значение, поскольку ссылка на A4 немедленно заменяется текстовой строкой B4. Следовательно, формула обрабатывается как =B4, что дает нам 6. По аналогии, если ввести в ячейке C5 формулу =ДВССЫЛ(A5), Excel вернет значение ячейки B5, то есть 9.

C4	:	X	✓	f _x	=ДВССЫЛ(A4)
	A	B	C	D	
1					
2					
3		Значение	Косвенная ссылка		
4	B4	6	6		
5	B5	9	9		
6					

Рис. 1. Простой пример функции ДВССЫЛ

1. Формулы моей электронной таблицы часто содержат ссылки на ячейки, диапазоны или на то и на другое вместе. Можно ли не изменять ссылки в формулах, а поместить их в отдельные ячейки, чтобы редактировать ссылки на ячейки и диапазоны, не затрагивая при этом формулы?

В диапазоне ячеек B4:H16 (рис. 2) приведены данные о ежемесячных продажах шести товаров за 12 месяцев. Сейчас я подсчитаю общие продажи каждого товара за месяцы со 2 по 12. Простейший способ подсчитать это — скопировать формулу СУММ(C6:C16) из ячейки C18 в диапазон D18:H18. Предположим, вам потребовалось изменить месяцы, по которым производится подсчет. Скажем, вы решите подсчитать общие продажи за месяцы 3–12. Можно изменить формулу в ячейке C18 на СУММ(C7:C16) и затем скопировать ее в диапазон D18:H18. Однако может быть не всегда удобно, поскольку вам придется копировать формулу из ячейки C18 в диапазон D18:H18, и, не просматривая формул, никто не узнает, какие строки суммируются.

C18	:	X	✓	f _x	=СУММ(ДВССЫЛ(C\$3&\$D\$2&":"&C\$3&\$E\$2))				
	A	B	C	D	E	F	G	H	I
1			Нижняя граница	Верхняя граница					
2			6	16					
3			C	D	E	F	G	H	
4		Месяц	Товар 1	Товар 2	Товар 3	Товар 4	Товар 5	Товар 6	
5		1	28	86	79	31	84	58	
6		2	38	7	61	1	20	2	
7		3	91	48	73	8	80	14	
8		4	33	32	24	77	29	80	
9		5	82	70	41	29	57	90	
10		6	75	40	15	92	55	91	
11		7	52	21	26	45	59	21	
12		8	19	6	35	67	40	81	
13		9	11	18	68	11	52	78	
14		10	90	30	52	32	30	1	
15		11	47	86	46	0	38	55	
16		12	69	71	75	65	53	52	
17									
18		Итого	607	429	516	427	513	565	
19									

Рис. 2. Функция ДВССЫЛ позволяет изменять ссылки на ячейки в формулах, не изменяя сами формулы

¹ При написании заметки использованы материалы книги Уэйн Л. Винстон. Microsoft Excel. Анализ данных и построение бизнес-моделей, глава 22.

Функция ДВССЫЛ предлагает другое решение. Я указал в ячейках D2 и E2 номера начальной и конечной суммируемых строк. Теперь, при использовании функции ДВССЫЛ, мне достаточно изменить значения в ячейках D2 и E2, чтобы конечная сумма обновилась, включив только те строки, которые мы хотим. Кроме того, значения ячеек D2 и E2 наглядно показывают, какие строки (месяцы) суммируются! Все, что мне требуется, — скопировать из ячейки C18 в диапазон D18:H18 формулу =СУММ(ДВССЫЛ(C\$3&\$D\$2&":"&C\$3&\$E\$2)). Каждая ссылка на ячейку в этой формуле обрабатывается как содержимое соответствующей ячейки. C\$3 обрабатывается как C, \$D\$2 — как 6, а \$E\$2 — как 16. Используя символ конкатенации — & (сцепления), Excel обрабатывает эту формулу как СУММ(C6:C16), что нам и требуется. Формула в ячейке D18 обрабатывается как СУММ(D6:D16), что также дает нужный нам результат. Конечно, если нам захочется просуммировать продажи, скажем, с 4 по 6 месяц, мы просто введем 8 в ячейку D2 и 10 в ячейку E2. После этого формула в ячейке C18 вернет $33 + 82 + 75 = 190$.

2. В ячейке B1 книги Excel, начиная с Лист1 и заканчивая Лист7 (рис. 3) содержатся данные о продажах товара за месяц. Есть ли какой-нибудь простой способ написать и скопировать формулу, которая выводила бы данные о продажах этого товара за каждый месяц на одном листе?

	A	B	C	D	E	F	G
1		1					
2							
3		N листа	Значение ячейки D1				
4	Лист	1	1				
5		2	4				
6		3	8				
7		4	10				
8		5	0				
9		6	15				
10		7	2				

Рис. 3. Данные о продажах товара за 1–7 месяцев, выведенные с помощью функции ДВССЫЛ

Предположим, Лист1 содержит данные о продажах за первый месяц, Лист2 — за второй и т.д. Пусть в первом месяце продажи равны 1. Например, вы хотите вывести продажи за все месяцы на одном листе. Нудный способ — подсчитать продажи за первый месяц с помощью формулы =Лист1!B1, продажи за второй месяц с помощью формулы =Лист2!B1 и т.д. Если ваши данные охватывают 100 месяцев, такое решение грозит грандиозной головной болью. Гораздо более изящный способ — вывести данные о продажах за первый месяц в ячейке C4 листа Лист1 с помощью формулы =ДВССЫЛ(\$A\$4&B4&!B1"), Excel обработает \$A\$4 как «Лист», B4 — как 1, и "!B1" — как строку текста !B1. Формула целиком будет обработана как =Лист1!B1, то есть покажет данные о продажах за первый месяц, содержащиеся в ячейке B1 листа Лист1. Скопировав эту формулу в диапазон C5:C10, вы отобразите содержимое ячейки B1 листов со 2 по 7. Обратите внимание: при копировании формулы из ячейки C4 в ячейку C5 ссылка на B4 заменяется ссылкой на B5, и формула в ячейке C5 возвращает значение ячейки Лист2!B1 и т.д.

3. Предположим, я суммирую значения из диапазона A5:A10 посредством формулы СУММ(A5:A10). Если вставить где-нибудь между 5 и 10 строками пустую строку, формула автоматически изменится на СУММ(A5:A11). Можно ли написать формулу, которая при вставке пустой строки между 5 и 10 строками все равно суммировала бы значения из диапазона A5:A10?

Рис. 4 иллюстрирует несколько способов сложения чисел из диапазона A5:A10. В ячейке A12 я ввел обычную формулу СУММ(A5:A10). Аналогичным образом формула СУММ(\$A\$5:\$A\$10) в ячейке C9 тоже возвращает значение 33. Тем не менее, если вставить строку между 5 и 10 строками, обе формулы попытаются сложить ячейки диапазона A5:A11.

Функция ДВССЫЛ (INDIRECT) предоставляет, как минимум, два способа сложения значений из диапазона A5:A10. В ячейке D9 я ввел формулу =СУММ(ДВССЫЛ("A5:A10")). Excel обрабатывает ДВССЫЛ("A5:A10") как строку текста "A5:A10", и поэтому, даже если я добавлю строку в электронную таблицу, формула по-прежнему будет суммировать значения из диапазона A5:A10.

	A	B	C	D
1				
2			Начало	Конец
3			5	10
4			Косвенный способ	
5	6		33	
6	7		=СУММ(ДВССЫЛ("A"&C3&":A"&D3))	
7	8			
8	9		Абсолютная ссылка	Косвенная ссылка
9	1		33	33
10	2		=СУММ(\$A\$5:\$A\$10)	=СУММ(ДВССЫЛ("A5:A10"))
11	Старый способ			
12	33		Вставьте пустую строку и посмотрите, что произойдет	
13	=СУММ(A5:A10)			

Рис. 4. Несколько способов сложения значений ячеек из диапазона A5:A10; под значением суммы написана формула

Еще один вариант сложения значений из диапазона A5:A10 при помощи функции ДВССЫЛ – формула =СУММ(ДВССЫЛ("A"&C3&":A"&D3)), которую я ввел в ячейке C5. Excel обрабатывает ссылку на C3 как 5, а ссылку на D3 — как 10, в результате чего формула преобразуется в СУММ(A5:A10). Вставка пустой строки между 5 и 10 строками никак не скажется на формуле, поскольку ссылка на C3 по-прежнему будет обрабатываться как 5, а ссылка на D3 — как 10. На рис. 5 показаны результаты суммирования, выполненные посредством наших четырех формул после того, как ниже 7-й строки была добавлена пустая строка.

	A	B	C	D
1				
2			Начало	Конец
3			5	10
4			Косвенный способ	
5	6		31	
6	7		=СУММ(ДВССЫЛ("A"&C3&":A"&D3))	
7	8			
8				
9	9		Абсолютная ссылка	Косвенная ссылка
10	1		33	31
11	2		=СУММ(\$A\$5:\$A\$11)	=СУММ(ДВССЫЛ("A5:A10"))
12	Старый способ			
13	33			
14	=СУММ(A5:A11)			

Рис. 5. Результаты, возвращенные формулами СУММ после того, как ниже строки 7 была вставлена пустая строка

Обратите внимание: классические формулы СУММ, не включающие оператор ДВССЫЛ, автоматически изменились и суммируют значения из диапазона A5:A11, по-прежнему возвращая значение 33. Две формулы СУММ, включающие оператор ДВССЫЛ, суммируют значения из диапазона A5:A10, в результате при вычислениях теряется число 2 (оно теперь находится в ячейке A11). Формулы СУММ, содержащие оператор ДВССЫЛ, возвращают значение 31.

